# 25 Years of Othello

### **Valentín Albillo (#1075, PPC #4747)**

Time flies like an arrow ('and fruit flies like a banana', as the late Groucho Marx once said). You sit for a while to enjoy a moment's remembrance or two and then you realize that 25 years have passed by since you finally laid your hands in your very own, brand-new, ultra-expensive tool called 'HP-41C'.

Back then I was a 'poor' student and I had to sell my beloved HP-67 to help pay for the new object of my desire. I resented very much being forced to part with my HP-67 and still to this day haven't fully recovered from the trauma, but the sheer temptation of that alphanumeric LCD display was too much to resist, specially since I was lent one way before I actually could afford it myself. Then, once the beast was in my hands, I duly indulged in the extreme pleasure, that only another HP calc lover can fully fathom, of reading the manuals down to the last iota, while trying each and every example and code snippet featured in there in the machine proper, though I must confess I was somewhat disdainful of the tiny Standar Pack, which seemed too meager and trivial to me, specially after having experienced the much better equivalents for both the HP-25 and the HP-67.

Next, I longed for a RAM module or two, then three, and soon afterwards I got my own (also very expensive) card reader and was lent a printer. And by then I had also become a proud PPC member (#4747 !) and made a lot of exciting discoveries in the realm of what later would be called HP-41C 'bugs' and 'synthetic programming'. Matter of fact, that was almost unavoidable as my recently acquired HP-41C had all bugs (un)known to man, so the first time I tried to SF IND 00 with 55 in R00, it worked as (un)expected; and storing values with STO IND 00 and 999 in R00 also did (!), albeit placing weird thing into program memory, such as an out-of-this-world STO M !! Shock ! Horror ! "May you live interesting times" says an old proverb. Indeed !

The few next months passed by in a frenzy, mastering the machine and developing all kinds of complex programs for it, mainly mathematical and engineering ones, with a good helping of synthetic utilities as well. It was soon apparent that, though the 41C's speed and RAM were adequate for those tasks, as well as some relatively simple games, they were ultimately quite limiting to even contemplate the possibility of dealing with substantially more complex games, say most typical board games such as chess, checkers, and the like. But I had been reading Martin Gardner's books regularly, where he once commented on an ancient game called Reversi, which though interesting and challenging to play, was relatively unknown. The rules were simple to grasp, the game was difficult to master, and I became fond of it. By the time I got the 41C, it was commercially available under the name "Othello". So, once I felt confident on my abilities, and having my 41C with 3 RAM modules plus printer, I decided to give it a try. Thus 41C's **Othello** was born.

DATAFILE Vxx Nx Page 1

#### Enter 41C's Othello

After some experimentation I decided that the game (including a nice print board routine) could indeed fit in that amount of RAM, thus providing an excellent demo for both printer and 41C capabilities. Given the speed limitations of the 41C, I refrained from minimax, alpha-beta search strategies and settled instead for a simple, fast positional strategy. The resulting 291-step program did play decently and would frequently surprise and even win against beginners and casual players.

I promptly showed it to my friends who were quite astonished, and demonstrated it at a major trade show where I was in charge of HP's calculators, contributing to a number of 41C sales among attendants. Next, I wrote an article and submitted it to PPC for its eventual publication. I was sure it would cause a great impression at the Club and among members once published, as it was very challenging to play, ideal to show off the 41C, and nothing like it existed or was deemed possible at the time.

### Othello goes down (under)

But lo and behold, it was *never* published! Mr. Nelson declined its publication for whatever reasons, and never wrote or contacted me with any hints about why it wasn't published, or any kind of request for me to reformat or rewrite the article. Frankly, I couldn't believe it, but I thought that he was certainly entitled to make whatever editorial choices he cared, perhaps he simply didn't like it or needed the space for more worthwhile materials or whatever.

That being so, and as an increasing number of materials I sent were ignored either, I had the fortune of getting in touch with John McGechie, then at Monash University down under in Australia, and promptly was friends with him. He told me about PPC Melbourne Chapter and their publication, PPC Technical Notes, and I sent them my 'rejected' materials, **Othello** among them. It was duly published in the September, 1980 issue of PPC TN, and generated a vast amount of interest.


### Othello goes up

So much so, in fact, that PPC Melbourne members started to disseminate it, to the point where I got a letter from HP's mythical Corvallis Division (see below), stating that they had received a copy ("The program has generated a lot of interest in our division") and kindly requesting that I submit it to the 41C User's Library. Needless to say, being so young, I was absolutely flattered that a program of mine had "generated a lot of interest" among the "gods" at Corvallis, and duly obligued by submitting it to the 41C User's Library, where it was accepted and renamed to "Reversi" for copyright reasons. HP then sent me an extremely kind acceptance letter (see below), enclosing a special *Certificate for Outstanding Contributions* redeemable for any one item among a list of 41C's hardware and accesories.

Why, I was flattered beyond belief, but I'll say for myself that I resisted temptation and did *not* redeem the Certificate, which I've kept to this day, as a wonderful reminder of those exhilarating days. Later, HP contacted me again to request my permission to publish **Othello** in the new *HP-41C Solutions Book Games II*. Sadly, my **Othello** didn't finally make it to the book but another version appeared

Page 2 DATAFILE Vxx Nx

instead, for whatever reasons. However, HP *did* include my 41C **Mini-Chess 5x5** in their superb Solutions Book, where it remains to this day for everyone to see.



#### Othello rules (Shouldn't that be "Othello's rules" instead? Oh, well, never mind)

Now, 25 years since my 41C **Othello** was brought into existence, I decided I would commemorate it by telling its history and reincarnating it for the occasion, essentially the very same program but this time slightly revamped for the **HP-71B**. But first, a brief introduction to Othello's rules for those not in the know:

	1	2	3	4	5	6	7	8			Othello is played in an 8x8 board , the human
											player plays the white pieces (0), the machine
[	-	-	-	-	-	-	_	-	]	1	plays the black pieces (#).
[	-	-	_	-	-	-	_	-	]	2	The initial start position can be either the
[	_	_	_	_	_	_	_	_	1	3	Diagonal Opening ( see left ) or the Parallel
[	_	_	_	0	#	_	_	_	1	4	Opening (both pieces in the same row instead
[	_	_	_	#	0	_	_	_	1	5	of diagonally placed).
Г	_	_	_	_	_	_	_	_	1	6	Both players take turns to make a move until
[	_	_	_	_	_	_	_	_	]	7	the game ends, either when the board is full,
_ [	_	_	_	_	_	_	_	_	]	8	or else if no player can make a legal move.
											To make a move, the player places one of his
	1	2	3	4	5	6	7	8			pieces in an empty location (-), subject to:
											- it <i>must</i> be adjacent to an enemy piece
	1	2	3	4	5	6	7	8			- at least one enemy piece must be enclosed
											between the just placed piece and another
Г	#	#	#	#	#	#	#	#	1	1	friendly piece (see Usage for passing).
											This is , any number of enemy pieces enclosed
											between the played piece and another friendly
_									_		piece are flipped over: they become friendly.
											No empty locations can be enclosed, only full
											rows of enemy pieces can be flipped. The rows
_									_		can be placed in any direction: horizontally,
											vertically or diagonally. If several rows are
										Ü	enclosed simultaneously all are flipped over.
											The one with the most pieces at the end wins.
	_	~	9	-	J	U	,	J			THE OHE WITH THE WORL PLECED OF THE END WIND.
		1 1 # # # # # # # # # # # # # # # # # #	1 2 1 2	1 2 3 1 2 3 1 2 3 1 4 # # 1 0 # # 1 # 0 0 1 # 0 # 1 # 0 #	1 2 3 4 1 2 3 4 1 2 3 4 1 2 0 4 1 0 0 0 1	1 2 3 4 5 1 2 3 4 5 1 2 3 4 5 1 2 3 4 5 1 2 3 4 5 1 2 3 4 5 1 3 4 5	1 2 3 4 5 6  1 2 3 4 5 6  1 2 3 4 5 6  1 4 # # # # #  0 # # # O O O  # O O O # #  # O # O O # #  # O # O	1 2 3 4 5 6 7  1 2 3 4 5 6 7  1 2 3 4 5 6 7  1 4 # # # # # #  0 # # # 0 0 0 0  # 0 0 0 # # #  # 0 # 0	1 2 3 4 5 6 7 8  1 2 3 4 5 6 7 8  1 2 3 4 5 6 7 8  1 2 3 4 5 6 7 8  1 4 # # # # # # # #  0 # # # 0 0 0 0 #  # 0 0 0 # # # #	1 2 3 4 5 6 7 8  1 2 3 4 5 6 7 8  1 2 3 4 5 6 7 8  1 2 3 4 5 6 7 8  1 4 # # # # # # # # # # # # # # # # # #	1 2 3 4 5 6 7 8  1 2 3 4 5 6 7 8  1 2 3 4 5 6 7 8  1 2 3 4 5 6 7 8  1 4 # # # # # # # ] 1  1 0 # # # 0 0 0 0 ] 2  # # 0 0 0 # # # ] 4  # 0 0 0 # # # ] 5  # 0 0 # # # ] 5  # 0 0 0 # # # ] 6  # # # # # # # ] 8

An example should make it clear: in the position above, if Black plays <u>52</u> it will flip <u>7</u> White pieces over (at 42, 43, 34, 25, 53, 54, and 62). White has no move.

DATAFILE Vxx Nx Page 3

# **Program listing**

**OTHELLO** for the HP-71B<sup>1</sup> is a short, 43-line program just under 2 Kb, which is a faithful conversion of the original for the 41C with the following enhancements:

- It'll check your pass for legality and, if illegal, will hint a legal move for you
- If neither player can move, the game ends, computing the final score and result
- The strategy is randomized at the same-merit level so games will never repeat

### **'OTHELLO'** (2,045 bytes)

```
1 DESTROY ALL @ OPTION BASE 0 @ STD @ RESTORE @ A$="
                                                          1 2 3 4 5 6 7 8"
 2 INTEGER B(99),M(7) @ DIM C$[60] @ DISP "** OTHELLO by VA **"
 3 DISP "Initializing" @ GOSUB 22 @ F=0 @ G=0 @ W$="" @ IF NOT V THEN 16
 4 ON ERROR GOTO 4 @ INPUT "Move (11-88)=", W$; W$ @ IF W$="" THEN 14
 5 S=ABS(INT(VAL(W$))) @ OFF ERROR @ X=S DIV 10 @ Y=MOD(S,10)
 6 IF X<1 OR X>8 OR Y<1 OR Y>8 THEN DISP "Illegal move!" @ W$="" @ GOTO 4
 7 DISP "Checking move" @ A=1 @ R=0 @ GOSUB 9 @ IF N=0 THEN X=0 @ GOTO 6
 8 DISP "You play"; S; ", flip"; N @ Q=Q+1 @ F=0 @ GOSUB 42 @ GOTO 16
 9 N=0 @ IF ABS(B(S)) THEN RETURN
10 FOR I=0 TO 7 @ E=M(I) @ H=S+E @ IF B(H)<>A THEN 13
11 H=H+E @ IF B(H)=A THEN 11 ELSE IF B(H)<>-A THEN 13
12 H=H-E @ IF H=S THEN 13 ELSE N=N+1 @IF R THEN RETURN ELSE B(H)=-A @ GOTO 12
13 NEXT I @ IF N THEN B(S)=-A @ H=S @ RETURN ELSE RETURN
14 GOSUB 29 @ IF N THEN DISP "No, you can play"; S @ W$=STR$(S) @ GOTO 4
15 IF F THEN 21 ELSE F=1
16 DISP "My turn. Thinking .. " @ K=1+12*(G<4)
17 S=NUM(C$[K]) @ IF ABS(B(S)) THEN C$[K,K]="" @ GOTO 20
18 A=-1 @ R=0 @ GOSUB 9 @ IF NOT N THEN K=K+1 @ GOTO 20 ELSE P=P+1 @ F=0
19 DISP "I play"; S; ", flip"; N @ GOSUB 42 @ C$[K,K]="" @ W$="" @ GOTO 4
20 IF K<=LEN(C$) THEN 17 ELSE DISP "I have no moves, pass" @ BEEP
21 IF F THEN DISP "Neither can play!" @ GOTO 43 ELSE F=1 @ W$="" @ GOTO 4
22 INPUT "Opening (D,P): ","D";R$ @ U=R$="D" @ INPUT "You 1st (Y,N): ","Y";R$
23 V=R$="Y" @ INPUT "P.board (Y,N): ","Y";R$ @ Z=R$#"Y" @ DELAY Z,0
24 FOR I=1 TO 60 @ READ B(I) @ NEXT I @ N=4 @ X=12 @ GOSUB 33 @ X=56
25 GOSUB 33 @ N=8 @ X=4 @ GOSUB 33 @ X=16 @ FOR G=1 TO 5 @ GOSUB 33 @ NEXT G
26 FOR I=1 TO 60 @ C$=C$&CHR$(B(I)) @ B(I)=0 @ NEXT I
27 B(44)=-1 @ B(54)=1 @ B(45+10*U)=-1 @ B(55-10*U)=1 @ READ M @ P=2 @ Q=2
28 GOSUB 35 @ DISP "Score: You:";Q;", Me:";P @ RETURN
29 DISP "Checking your pass .. " @ K=LEN(C$)
30 S=NUM(C$[K]) @ IF ABS(B(S)) THEN C$[K,K]="" @ K=MIN(K,LEN(C$)) @ GOTO 32
31 A=1 @ R=1 @ GOSUB 9 @ IF N THEN RETURN ELSE K=K-1
32 IF K THEN 30 ELSE RETURN
33 FOR I=1 TO N @ P=INT(RND*N+1)+X @ Q=INT(RND*N+1)+X @ E=B(P) @ B(P)=B(Q)
34 B(Q)=E @ NEXT I @ X=X+N @ RETURN
35 IF Z THEN RETURN ELSE DISP USING '/,21A,/,3X,19"-"';A$
36 FOR H=1 TO 8 @ DISP H;"["; @ FOR I=1 TO 8 @ U=B(10*H+I)
37 DISP " ";CHR$(45-10*(U>0)+34*(U<0)); @ NEXT I @ DISP " ]";H @ NEXT H
38 DISP USING '3X,19"-",/,21A,/';A$ @ RETURN
39 DATA 11,18,81,88,13,16,61,68,31,38,83,86,63,66,33,36,14,15,51,58,41,48,84
40 DATA 85,64,65,53,56,43,46,34,35,74,75,52,57,42,47,24,25,73,76,62,67,32,37
41 DATA 23,26,12,17,71,78,21,28,82,87,72,77,22,27,1,9,10,11,-1,-9,-10,-11
42 P=P-A*N @ Q=Q+A*N @ G=G+1 @ GOSUB 28 @ IF P+Q<>64 AND P*Q<>0 THEN RETURN
43 IF P>Q THEN DISP "I WIN" ELSE IF P=Q THEN DISP "A tie" ELSE DISP "You win"
```

Page 4 DATAFILE Vxx Nx

<sup>&</sup>lt;sup>2</sup> No HP-71? No problem. Google the web for **Emu71**, a free emulator for Windows, and/or **HP-71X**, another outstanding emulator, this time for your HP48/49.

## **Usage instructions** (see also 'Sample game' below)

• To start the program, press:

```
>RUN

** OTHELLO by VA **
Initializing
Opening (D,P): D
```

• To select Diagonal Opening, press: [ENTER]

To select Parallel Opening, press: P [ENTER]

```
You 1st (Y,N): Y
```

• To have first move press: [ENTER], else press: N [ENTER]

```
P.board (Y,N): Y
```

- To print the board after every move press: [ENTER], else: N [ENTER]
- Whatever the turn, the actual score is shown:

```
Score: You: nn , Me: mm
```

• When it's your turn to move, you'll be prompted:

```
Move (11-88)=
```

enter the coordinates of the square where you move to (11 to 88) and [ENTER]

• Your move will be checked for legality and, if not legal, you'll be warned:

```
Illegal move!
```

and then you'll be prompted again to enter a legal move.

• Should you be unable to make a legal move, when prompted just press:

```
[ENTER]
```

to <u>pass</u>. The program will check whether your pass is legal or not:

```
Checking your pass ..
```

and if not, it will *deny* your pass, showing one legal move you *could* make:

```
No, you can play 46
```

prompting you to enter your move again, with the suggested move as a *default* which you can accept by pressing [ENTER], or else key in another instead.

• Once you make a legal move, it is duly acknowledged and scored:

```
You play 46 , flip 1
```

then the updated board is optionally printed, and the program thinks its move:

```
My turn. Thinking ..

I play 36 , flip 1 or I have no moves, pass
```

• When the game's over, the final score is printed, and the result is given:

```
I WIN Or A tie Or You win
```

DATAFILE Vxx Nx Page 5

### Sample game

This is a commented sample game, partially shown. Since the program uses a randomized strategy, you'll generally never get to play the same game twice, so the program's moves may be different to what's shown here. In this particular game, the program won convincingly, but you certainly might do a lot better!

**Caveat emptor:** Should you decide to illegally pass with the intention of then accepting by default the hint the program will kindly offer, be warned that the program's "suggested" move, while perfectly legal, it's nevertheless the *very worse* it could find for your side! So accept it at your own risk.

```
>RUN
** OTHELLO by VA **
Initializing
Opening (D,P): D [ENTER]
                           {we chose Diagonal Opening}
You 1st (Y,N): Y
                           {we'll play first}
                 [ENTER]
P.board (Y,N): Y
                 [ENTER]
                           {we want the board printed after
                            every move, at some HP-IL device }
     1 2 3 4 5 6 7 8
                           {the initial position is printed.
                            Normally you'll chose printing if
                            you have some print-capable HP-IL
 2 [ - - - - - - -
                            device, such as a real printer or
 3 [ - - - - - 1 3
                            a 24 rows x 80 columns display.
 4 [ - - - 0 # - - - ] 4
                            For instance, Emu71 provides some
 5 [ - - - # 0 - - - ] 5
                            virtual HP-IL display that can be
                            used to print boards and messages
 7 [ - - - - - - -
                            with this program. If you've got
                            no printing device and specify so
                            when asked, you'll get a 1-second
     1 2 3 4 5 6 7 8
                            delay after every message instead}
Score: You: 2 , Me: 2
                           {the initial score: 2 pieces each}
Move (11-88)=64 [ENTER]
                           {we place a piece at 64}
Checking move
                           {it checks our move for legality}
                           {legal, flips Black's piece at 54}
You play 64 , flip 1
                           {prints new position after our move}
     1 2 3 4 5 6 7 8
 3 [ - - - - - -
 4 [ - - - 0 # - - -
 5 [ - - - 0 0 - - -
 6 [ - - - 0 - - - ] 6
 8 [ - - - - - ] 8
   -----
     1 2 3 4 5 6 7 8
```

Page 6 DATAFILE Vxx Nx

```
Score: You: 4 , Me: 1
                           {the new score after our move}
My turn. Thinking ..
                           {now the program computes its move}
I play 63 , flip 1
                           {plays at 63 flipping a piece back}
     1 2 3 4 5 6 7 8
                            {prints new position after its move}
                            {notice that once placed, pieces do
 1 [ - - - - - ] 1
                            never move, they can be only be
 3 [ - - - - - 1 3
                            flipped over back and forth any
 4 [ - - - 0 # - - - ] 4
                            number of times but always remain
 5 [ - - - # 0 - - - ] 5
                            where first placed. Thus the board
 6 [ - - # 0 - - - ] 6
                            gets fuller and fuller with every
                            non-passing move, and the game ends
 7 [ - - - - - - ] 7
 8 [ - - - - - 1 8
                            when it is completely full or else
                            if no side can make a legal move }
    -----
     1 2 3 4 5 6 7 8
Score: You: 3 , Me: 3
                           {the new score after its move}
Move (11-88) =
                           {the game goes on ...}
My turn. Thinking ...
I play 83 , flip 2
     1 2 3 4 5 6 7 8
 1 [ # # # # # # # ] 1
 2 [ 0 0 # # # 0 # # ] 2
 3 [ 0 0 # # # # # # ] 3
 4 [ 0 # 0 0 # 0 - # ] 4
 5 [ 0 0 0 0 # # # # ] 5
 6 [ # 0 # # 0 # # # ] 6
 7 [ - # - - - # # # ] 7
 8 [ # # # - # # 0 # ] 8
     1 2 3 4 5 6 7 8
Score: You: 16 , Me: 42
                          {we attempt to illegally pass}
Move (11-88) = [ENTER]
                           {the program checks our pass}
Checking your pass ...
No, you can play 71
                           {not legal, we have at least 1 move}
Move (11-88)=71 [ENTER]
                           {we accept the suggested legal move}
                           {the game goes on ...}
     1 2 3 4 5 6 7 8
 1 [ # # # # # # # ] 1
                           {notice that we did try to pass when
 2 [ 0 0 # # # 0 # # ] 2
                           actually there were legal moves to
 3 [ 0 0 # # # # # # ] 3
                           play; the program detects this and
 4 [ 0 # 0 0 # # # # ] 4
                           denies the attempted pass, while
 5 [ 0 0 0 0 # # # # ] 5
                           also offering a legal move for your
 6 [ 0 0 0 0 # # # # ] 6
                           side. This can be used to get a hint
 7 [ 0 0 0 0 0 # # # ] 7
                           when you're not sure whether you can
 8 [ # # # - # # 0 # ] 8
                           play or not, and also to make the
                           program play against itself !}
      ______
```

DATAFILE Vxx Nx Page 7

1 2 3 4 5 6 7 8

```
Score: You: 22 , Me: 41
                           {the game is about to end}
                           {the program computes its move}
My turn. Thinking ..
                           {forced, flipping 5 of your discs}
I play 84, flip 5
     1 2 3 4 5 6 7 8
                           {the final position is printed}
 1 [ # # # # # # # ] 1
                           {the program notices that the board
 2 [ 0 0 # # # 0 # # ] 2
                            is choke full and proceeds to make
 3 [ 0 0 # # # # # # ] 3
                            the final piece tally, which in
 4 [ 0 # 0 # # # # # ] 4
                            this particular sample game reveals
                            a most humiliating defeat for the
 5 [ 0 0 0 # # # # # ] 5
                            human side }
 6 [ 0 0 0 # # # # # ] 6
 7 [ 0 0 0 # # # # # ] 7
 8 [ # # # # # 0 # ] 8
     1 2 3 4 5 6 7 8
                            {the final score, 47/17 for black}
Score: You: 17 , Me: 47
I WIN
                            {the program won}
```

#### Final remarks

It's certainly fun to write any sort of game playing program, and for those of us mathematically oriented, classic non-random board games such as Chess, Checkers and Othello do hold an special fascination. 25 years ago, a much younger version of myself was absolutely delighted with its new acquisition, an expensive, state-of-the-art handheld computing device called an HP-41C, and wished to explore how far he could go with this fabulous machine's capabilities. Implementing **Othello** in this supposedly limited hardware provided an extremely exciting challenge and was the source of many satisfactory experiences, probably strengthening an incipient vocation for all things related to programming in general.

After succeeding with the 41C, I took the habit of writing an exact translation of this particular version of **Othello** for every worthwhile small machine I laid my hands on, so versions came into existence for most non-graphical SHARP handhelds (such as the venerable PC-1211), as well as full-graphic versions for the graphics models (such as the SHARP PC-1350). I also created versions for very small microcomputers such as the HP-85 and Sinclair ZX-81, to name a few (the version for the latter was compiled to Z80 machine code, so it ran at absolutely blinding speed). There's also the HP-71B version specially created for this article.

It always was an interesting and normally quick and easy challenge (the difficulty was in writing the *original* version, mind you), and I was pretty satisfied to implement **Othello** in newer machines, even ones that weren't my own. As for the 41C, I later did write an 8x8 Checkers program and the aforementioned 5x5 MiniChess, among many other games. But, perhaps due to purely sentimental reasons playing no small part, **Othello's** always been my favourite and always will be. Hope you've enjoyed my sharing these fonds memories of mine with you all.

Page 8 DATAFILE Vxx Nx